A symbol conflict arises when a package operation would result in two distinct symbols with the same name being accessible in a single package. There are exactly four ways this can happen.

Inheriting: When you use (via use-package or the defpackage :use clause) a package that exports a distinct symbol sharing the same name with a symbol already accessible in the using package. The conflict can arise between two inherited symbols, or between an inherited symbol and a present symbol.

Importing: When you import a symbol and a distinct symbol with the same name is already accessible in the target package.

Exporting: When you export a symbol from a package and a distinct symbol with the same name is already accessible in any package that uses that package.

Uninterning: When you unintern a shadowing symbol and the absence of its shadowing results in two distinct symbols with the same name being accessible via inheritance.

Here are short examples of each case. They all assume an environment in which this package definition is in effect:

(defpackage #:auto (:documentation "An automotive package.") (:use) (:intern #:tan) (:export #:car))

Inheritance conflict:

(defpackage #:inheritance-conflict (:use #:cl #:auto)) => USE-PACKAGE #[PACKAGE "COMMON-LISP"] causes name-conflicts in #[PACKAGE "INHERITANCE-CONFLICT"] between the following symbols: AUTO:CAR, COMMON-LISP:CAR [Condition of type NAME-CONFLICT] See also: Common Lisp Hyperspec, 11.1.1.2.5 [:section]

Import conflict:

(defpackage #:import-conflict (:use #:cl)) (import '(auto:car) '#:import-conflict) => IMPORT AUTO:CAR causes name-conflicts in #[PACKAGE "IMPORT-CONFLICT"] between the following symbols: AUTO:CAR, COMMON-LISP:CAR [Condition of type NAME-CONFLICT] See also: Common Lisp Hyperspec, 11.1.1.2.5 [:section]

Export conflict:

(defpackage #:export-conflict (:use #:cl #:auto) ;; Avoid inheritance conflict with shadowing (:shadow #:car)) (export 'auto::tan '#:auto) => EXPORT AUTO::TAN causes name-conflicts in #[PACKAGE "EXPORT-CONFLICT"] between the following symbols: AUTO::TAN, COMMON-LISP:TAN [Condition of type NAME-CONFLICT] See also: Common Lisp Hyperspec, 11.1.1.2.5 [:section]

Unintern conflict:

(defpackage #:unintern-conflict (:use #:cl #:auto) ;; Avoid inheritance conflict with shadowing (:shadow #:car)) (unintern 'unintern-conflict::car '#:unintern-conflict) => UNINTERN UNINTERN-CONFLICT::CAR causes name-conflicts in #<PACKAGE "UNINTERN-CONFLICT"> between the following symbols: AUTO:CAR, COMMON-LISP:CAR [Condition of type NAME-CONFLICT] See also: Common Lisp Hyperspec, 11.1.1.2.5 [:section]

If you use Common Lisp, you have to interact with the package system. I think the best way to get along with the package system is not to keep it at arm’s length while holding your nose, but to learn how it works, inside and out.